
    PinStripe

A minimalist Stripe integration for Elixir.
Warning
This library is still experimental! It's thoroughly tested in ExUnit, but that's it.
Stripe doesn't provide an official Elixir SDK, and maintaining a full-featured SDK has proven to be a challenge. As I see it, this is because the Elixir community is pretty senior-driven. People have no problem rolling their own integration with the incredible Req library. 
This library is an attempt to wrap those community learnings in an easy-to-use package modeled after patterns set forth in Dashbit's SDKs with Req: Stripe article by Wojtek Mach. 
My hope is that this should suffice for 95% of all apps that need to integrate with Stripe, and that the remaining 5% of use cases have a built-in escape hatch with Req. 
Features
	Simple API Client built on Req with automatic ID prefix recognition
	Webhook Handler DSL using Spark for clean, declarative webhook handling
	Automatic Signature Verification for webhook security
	Code Generators powered by Igniter for zero-config setup
	Sync with Stripe to keep your local handlers in sync with your Stripe dashboard

Installation
Using Igniter (Recommended)
The fastest way to install is using the Igniter installer:
mix igniter.install pin_stripe

This will:
	Add the dependency to your mix.exs
	Replace Plug.Parsers with PinStripe.ParsersWithRawBody in your Phoenix endpoint
	Generate a StripeWebhookController with example event handlers
	Add the webhook route to your router (default: /webhooks/stripe)
	Configure webhook_paths in config/runtime.exs
	Add DSL formatting support to .formatter.exs

Then configure your Stripe credentials:
# config/runtime.exs
config :pin_stripe,
  stripe_api_key: System.get_env("STRIPE_SECRET_KEY"),
  stripe_webhook_secret: System.get_env("STRIPE_WEBHOOK_SECRET")
Manual Installation
Add to your mix.exs:
def deps do
  [
    {:pin_stripe, "~> 0.3"}
  ]
end
Then follow the Manual Setup instructions below.
Multiple Webhook Endpoints
To handle multiple webhook endpoints (e.g., regular and Stripe Connect):
	Add paths to config:

# config/runtime.exs
config :pin_stripe,
  webhook_paths: ["/webhooks/stripe", "/webhooks/stripe_connect"]
	Create additional controllers:

defmodule MyAppWeb.StripeConnectWebhookController do
  use PinStripe.WebhookController

  handle "account.updated", fn event ->
    # Handle Connect-specific events
    :ok
  end
end
	Add routes:

scope "/webhooks" do
  post "/stripe", MyAppWeb.StripeWebhookController, :create
  post "/stripe_connect", MyAppWeb.StripeConnectWebhookController, :create
end
Manual Installation (without Igniter)
	Add config:

# config/runtime.exs
config :pin_stripe,
  stripe_api_key: System.get_env("STRIPE_SECRET_KEY"),
  stripe_webhook_secret: System.get_env("STRIPE_WEBHOOK_SECRET"),
  webhook_paths: ["/webhooks/stripe"]
	Replace Plug.Parsers in your endpoint:

# lib/my_app_web/endpoint.ex
plug PinStripe.ParsersWithRawBody,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Jason
	Create a webhook controller:

# lib/my_app_web/stripe_webhook_controller.ex
defmodule MyAppWeb.StripeWebhookController do
  use PinStripe.WebhookController
end
	Add the route:

# lib/my_app_web/router.ex
scope "/webhooks" do
  post "/stripe", MyAppWeb.StripeWebhookController, :create
end
	Add formatter config:

# .formatter.exs
[
  import_deps: [:pin_stripe]
]
Handling Stripe Webhooks
PinStripe provides a clean DSL for handling webhook events. Webhooks are automatically verified, parsed, and dispatched to your handlers.
Function Handlers
For simple event processing:
defmodule MyAppWeb.StripeWebhookController do
  use PinStripe.WebhookController

  handle "customer.created", fn event ->
    customer_id = event["data"]["object"]["id"]
    email = event["data"]["object"]["email"]
    
    # Your business logic here
    MyApp.Customers.create_from_stripe(customer_id, email)
    
    :ok
  end

  handle "customer.updated", fn event ->
    # Handle customer updates
    :ok
  end

  handle "invoice.payment_succeeded", fn event ->
    # Handle successful payments
    :ok
  end
end
Module Handlers
For more complex event processing, use separate handler modules:
defmodule MyAppWeb.StripeWebhookController do
  use PinStripe.WebhookController

  handle "customer.subscription.created", MyApp.StripeWebhookHandlers.SubscriptionCreated
  handle "customer.subscription.updated", MyApp.StripeWebhookHandlers.SubscriptionUpdated
  handle "customer.subscription.deleted", MyApp.StripeWebhookHandlers.SubscriptionDeleted
end
defmodule MyApp.StripeWebhookHandlers.SubscriptionCreated do
  @moduledoc """
  Handles subscription creation events.
  """

  def handle_event(event) do
    subscription = event["data"]["object"]
    customer_id = subscription["customer"]
    
    # Complex business logic
    with {:ok, user} <- MyApp.Users.find_by_stripe_customer(customer_id),
         {:ok, _subscription} <- MyApp.Subscriptions.create(user, subscription) do
      :ok
    else
      {:error, reason} ->
        {:error, reason}
    end
  end
end
Generating Handlers
Quickly scaffold handlers:
mix pin_stripe.gen.handler customer.created
mix pin_stripe.gen.handler customer.subscription.created --handler-type module
mix pin_stripe.gen.handler charge.succeeded --handler-type module --module MyApp.Payments.ChargeHandler

Syncing with Stripe
Sync your local handlers with your Stripe webhook configuration:
mix pin_stripe.sync_webhook_handlers

This fetches your Stripe webhook endpoints, compares them with existing handlers, and generates stubs for missing events.
Options:
	--handler-type function|module|ask
	--skip-confirmation or -y
	--api-key or -k

Example output:
Fetching webhook endpoints from Stripe...

Found 1 webhook endpoint(s):
  • https://myapp.com/webhooks/stripe (5 events)

Collecting all enabled events...

Events configured in Stripe:
  ✓ customer.created (handler exists)
  ✗ customer.updated (missing)
  ✗ invoice.payment_succeeded (missing)
  ✓ subscription.created (handler exists)
  ✗ subscription.deleted (missing)

Found 3 missing handler(s) out of 5 total Stripe event(s).

Generate handlers for missing events? (y/n) y

What type of handlers would you like to generate?
1. function
2. module
3. ask
> 1

Generating handlers...
  • customer.updated (function handler)
  • invoice.payment_succeeded (function handler)
  • subscription.deleted (function handler)

✓ Done! Generated 3 new handler(s).
Calling the Stripe API
Simple CRUD interface built on Req:
alias PinStripe.Client

# Fetch a customer by ID
{:ok, response} = Client.read("cus_123")
customer = response.body

# List customers with pagination
{:ok, response} = Client.read(:customers, limit: 10, starting_after: "cus_123")
customers = response.body["data"]

# Create a customer
{:ok, response} = Client.create(:customers, %{
  email: "customer@example.com",
  name: "Jane Doe",
  metadata: %{user_id: "12345"}
})

# Update a customer
{:ok, response} = Client.update("cus_123", %{
  name: "Jane Smith",
  metadata: %{premium: true}
})

# Delete a customer
{:ok, response} = Client.delete("cus_123")
Automatic ID recognition:
Client.read("cus_123")      # => /customers/cus_123
Client.read("sub_456")      # => /subscriptions/sub_456
Client.read("price_789")    # => /prices/price_789
Client.read("product_abc")  # => /products/product_abc
Client.read("inv_xyz")      # => /invoices/inv_xyz
Client.read("evt_123")      # => /events/evt_123
Client.read("cs_test_abc")  # => /checkout/sessions/cs_test_abc
Entity types:
Client.create(:customers, %{email: "test@example.com"})
Client.create(:subscriptions, %{customer: "cus_123", items: [%{price: "price_abc"}]})
Client.create(:products, %{name: "Premium Plan"})
Client.create(:prices, %{product: "prod_123", unit_amount: 1000, currency: "usd"})
Client.create(:checkout_sessions, %{mode: "payment", line_items: [...]})

Client.read(:customers, limit: 100)
Client.read(:subscriptions, customer: "cus_123")
Client.read(:invoices, status: "paid")
Bang functions:
# Raises RuntimeError on failure
response = Client.read!("cus_123")
customer = Client.create!(:customers, %{email: "test@example.com"})
Advanced usage with Req:
# Direct Req request with custom options
{:ok, response} = Client.request("/charges/ch_123", retry: :transient)

# Or build a custom client
client = Client.new(receive_timeout: 30_000)
{:ok, response} = Req.get(client, url: "/customers/cus_123")
Testing
Test your Stripe integrations without making real API calls.
Mock Helpers
High-level helpers for stubbing Stripe API responses:
Setup:
# config/test.exs
config :pin_stripe,
  req_options: [plug: {Req.Test, PinStripe}]
Examples:
alias PinStripe.Test.Mock
alias PinStripe.Client

test "reads a customer" do
  Mock.stub_read("cus_123", %{
    "id" => "cus_123",
    "email" => "test@example.com"
  })
  
  {:ok, response} = Client.read("cus_123")
  assert response.body["email"] == "test@example.com"
end

test "lists customers" do
  Mock.stub_read(:customers, %{
    "object" => "list",
    "data" => [
      %{"id" => "cus_1", "email" => "user1@example.com"},
      %{"id" => "cus_2", "email" => "user2@example.com"}
    ],
    "has_more" => false
  })
  
  {:ok, response} = Client.read(:customers)
  assert length(response.body["data"]) == 2
end

test "creates a product" do
  Mock.stub_create(:products, %{
    "id" => "prod_new",
    "name" => "Test Product"
  })
  
  {:ok, response} = Client.create(:products, %{name: "Test Product"})
  assert response.body["id"] == "prod_new"
end

test "updates a customer" do
  Mock.stub_update("cus_123", %{
    "id" => "cus_123",
    "name" => "Updated Name"
  })
  
  {:ok, response} = Client.update("cus_123", %{name: "Updated Name"})
  assert response.body["name"] == "Updated Name"
end

test "deletes a customer" do
  Mock.stub_delete("cus_123", %{
    "id" => "cus_123",
    "deleted" => true,
    "object" => "customer"
  })
  
  {:ok, response} = Client.delete("cus_123")
  assert response.body["deleted"] == true
end

test "handles not found error" do
  Mock.stub_error("cus_nonexistent", 404, %{
    "error" => %{
      "type" => "invalid_request_error",
      "code" => "resource_missing"
    }
  })
  
  assert {:error, %{status: 404}} = Client.read("cus_nonexistent")
end

test "handles validation error on create" do
  Mock.stub_error(:customers, 400, %{
    "error" => %{
      "message" => "Invalid email address",
      "param" => "email"
    }
  })
  
  {:error, response} = Client.create(:customers, %{email: "invalid"})
  assert response.body["error"]["param"] == "email"
end

test "handles API key error for any request" do
  Mock.stub_error(:any, 401, %{
    "error" => %{"message" => "Invalid API key"}
  })
  
  assert {:error, %{status: 401}} = Client.read("cus_123")
end
Available helpers:
	stub_read/2 - Stub read operations (by ID or entity type for lists)
	stub_create/2 - Stub create operations (by entity type)
	stub_update/2 - Stub update operations (by ID)
	stub_delete/2 - Stub delete operations (by ID)
	stub_error/3 - Stub error responses (for ID, entity type, or :any)

These helpers work seamlessly with fixtures:
test "uses fixture with helper" do
  customer = PinStripe.Test.Fixtures.load(:customer)
  Mock.stub_read("cus_123", customer)
  
  {:ok, response} = Client.read("cus_123")
  assert response.body["object"] == "customer"
end

test "uses error fixture with helper" do
  error = PinStripe.Test.Fixtures.load(:error_404)
  Mock.stub_error("cus_missing", 404, error)
  
  assert {:error, %{status: 404}} = Client.read("cus_missing")
end
Advanced stubbing:
test "handles multiple operations in one stub" do
  Mock.stub(fn conn ->
    case {conn.method, conn.request_path} do
      {"GET", "/v1/customers/" <> id} ->
        Mock.json(conn, %{"id" => id, "email" => "#{id}@example.com"})
      
      {"POST", "/v1/customers"} ->
        Mock.json(conn, %{"id" => "cus_new", "email" => "new@example.com"})
      
      {"DELETE", "/v1/customers/" <> id} ->
        Mock.json(conn, %{"id" => id, "deleted" => true})
      
      _ ->
        conn
    end
  end)
  
  {:ok, read_resp} = Client.read("cus_123")
  {:ok, create_resp} = Client.create(:customers, %{email: "new@example.com"})
  {:ok, delete_resp} = Client.delete("cus_123")
end
See PinStripe.Test.Mock for full documentation.
Fixtures
Generate realistic test fixtures from actual Stripe data:
Two types:
	Error fixtures (atoms like :error_404): Instant, no setup required
	API resources (atoms like :customer): Require Stripe CLI, created once and cached

⚠️ API resource fixtures create real test data in your Stripe account. Commit generated fixtures to git.
Requirements for API resources:
	Stripe CLI installed
	Test mode API key

Setup:
# config/test.exs
config :pin_stripe,
  stripe_api_key: System.get_env("STRIPE_SECRET_KEY"),
  req_options: [plug: {Req.Test, PinStripe}]
Usage:
# Error fixtures - instant
error = PinStripe.Test.Fixtures.load(:error_404)

# API resources - created once, cached
customer = PinStripe.Test.Fixtures.load(:customer)
Customization:
customer = PinStripe.Test.Fixtures.load(:customer, email: "alice@test.com")
event = PinStripe.Test.Fixtures.load("customer.created", data: %{...})
API version management:
When you upgrade Stripe API versions:
mix pin_stripe.sync_api_version

Supported fixtures:
	API Resources: customer, product, price, subscription, invoice, charge, payment_intent, refund
	Webhook Events: customer.created, customer.subscription.updated, invoice.paid, etc.
	Errors: error_404, error_400, error_401, error_429

See PinStripe.Test.Fixtures for full documentation.
Configuration
# config/runtime.exs
config :pin_stripe,
  stripe_api_key: System.get_env("STRIPE_SECRET_KEY"),
  stripe_webhook_secret: System.get_env("STRIPE_WEBHOOK_SECRET"),
  webhook_paths: ["/webhooks/stripe"]

# config/test.exs  
config :pin_stripe,
  req_options: [plug: {Req.Test, PinStripe}]

Special Thanks

    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
[0.3.0] - 2025-12-19
Changed
⚠️ BREAKING CHANGE: Simplified webhook handler architecture
Webhook event handlers are now defined directly in your WebhookController instead of a separate WebhookHandler module. This reduces indirection and simplifies the overall architecture.
Before (0.2.x):
defmodule MyApp.StripeWebhookHandlers do
  use PinStripe.WebhookHandler
  
  handle "customer.created", fn event -> ... end
end

defmodule MyAppWeb.StripeWebhookController do
  use PinStripe.WebhookController,
    handler: MyApp.StripeWebhookHandlers
end
After (0.3.0):
defmodule MyAppWeb.StripeWebhookController do
  use PinStripe.WebhookController
  
  handle "customer.created", fn event -> ... end
end
Migration: Move your handle declarations from your WebhookHandler module into your WebhookController. Handler modules (for complex handlers) should remain in lib/my_app/stripe_webhook_handlers/ but are now referenced directly from the controller.
[0.2.2] - 2025-12-19
Added
	PinStripe.Test.Fixtures - Module for loading realistic Stripe test data	Error fixtures for all HTTP error codes (400, 401, 403, 404, 429, 500, etc.)
	API resource fixtures with caching support
	mix pin_stripe.sync_api_version task for fixture version management


	PinStripe.Test.Mock - High-level mocking helpers for Stripe API testing	stub_read/2 - Stub read operations by ID or entity type
	stub_create/2 - Stub create operations by entity type
	stub_update/2 - Stub update operations by ID
	stub_delete/2 - Stub delete operations by ID
	stub_error/3 - Stub error responses for any operation



[0.2.1] - 2025-12-19
Fixed
	Webhook handler generator bug: original implementation used text search to validate handler didn't already exist, which picked up examples in the @doc tag.

[0.2.0] - 2025-12-19
Changed
🎉 Complete rebrand from TinyElixirStripe to PinStripe!
The library has been rebranded with a new name that better reflects its sharp, professional approach to Stripe integration. Inspired by the pinstripe suit - clean, elegant, and professional.
	Package name: tiny_elixir_stripe → pin_stripe
	Module namespace: TinyElixirStripe → PinStripe
	Config atoms: :tiny_elixir_stripe → :pin_stripe
	Mix tasks: mix tiny_elixir_stripe.* → mix pin_stripe.*

This is a breaking change. If you're upgrading from TinyElixirStripe:
	Update your mix.exs dependency from :tiny_elixir_stripe to :pin_stripe
	Replace all TinyElixirStripe module references with PinStripe
	Update config keys from :tiny_elixir_stripe to :pin_stripe
	Update mix task calls (e.g., mix pin_stripe.install instead of mix tiny_elixir_stripe.install)

[0.0.2] - 2025-12-18
Added
	Docs with HexDocs
	This changelog
	Hex publish configurations

0.0.1 - 2025-12-18
Added
	Initial release
	Stripe API client built on Req with automatic ID prefix recognition
	Webhook handler DSL using Spark
	Automatic webhook signature verification
	mix tiny_elixir_stripe.install - Igniter-powered installation task
	mix tiny_elixir_stripe.gen.handler - Generate webhook event handlers
	mix tiny_elixir_stripe.sync_webhook_handlers - Sync handlers with Stripe dashboard
	mix tiny_elixir_stripe.update_supported_events - Update supported Stripe events list
	Support for both function and module-based webhook handlers
	CRUD operations for common Stripe entities (customers, subscriptions, products, prices, etc.)
	Comprehensive test coverage
	Full documentation and guides



  

    
PinStripe.WebhookHandler.Dsl.Handler 
    



      
Represents a webhook event handler.

      




  

    
PinStripe.WebhookHandler.Transformers.GenerateHandleEvent 
    



      
Transformer that generates handle_event/2 functions and Phoenix controller code
for webhook handlers.

        after?(_)

      


        Callback implementation for Spark.Dsl.Transformer.after?/1.



    


    
      
        after_compile?()

      


        Callback implementation for Spark.Dsl.Transformer.after_compile?/0.



    


    
      
        before?(_)

      


        Callback implementation for Spark.Dsl.Transformer.before?/1.



    


    
      
        transform(dsl_state)

      


        Callback implementation for Spark.Dsl.Transformer.transform/1.

Callback implementation for Spark.Dsl.Transformer.after?/1.

Callback implementation for Spark.Dsl.Transformer.after_compile?/0.

Callback implementation for Spark.Dsl.Transformer.before?/1.

Callback implementation for Spark.Dsl.Transformer.transform/1.

  


        

      


  

    
PinStripe 
    



      
A minimal Stripe SDK for Elixir.
Warning
This library is still experimental! It's thoroughly tested in ExUnit, but that's it.
PinStripe provides:
	Simple API Client - Built on Req with automatic ID prefix recognition
	Webhook Handler DSL - Using Spark for clean, declarative webhook handling
	Automatic Signature Verification - For webhook security
	Code Generators - Powered by Igniter for zero-config setup
	Sync with Stripe - Keep your local handlers in sync with your Stripe dashboard

Quick Start
Install with Igniter:
mix pin_stripe.install
Configure your API key:
config :pin_stripe,
  stripe_api_key: System.get_env("STRIPE_SECRET_KEY")
Handle webhook events:
defmodule MyApp.StripeWebhookHandlers do
  use PinStripe.WebhookHandler

  handle "customer.created", fn event ->
    # Your logic here
    :ok
  end
end
Call the Stripe API:
alias PinStripe.Client

{:ok, response} = Client.read("cus_123")
{:ok, response} = Client.create(:customers, %{email: "test@example.com"})
For full documentation, see the README.

      




  

    
PinStripe.Client 
    



      
A minimal Stripe API client built on Req.
This module provides a simple interface for interacting with the Stripe API,
covering the most common operations while maintaining an escape hatch to Req
for advanced use cases.
Configuration
Configure your Stripe API key in your application config:
config :pin_stripe,
  stripe_api_key: "sk_test_..."
For testing, configure the client to use Req.Test:
config :pin_stripe,
  req_options: [plug: {Req.Test, PinStripe}]
CRUD Operations
The client supports standard CRUD operations with a consistent API:
Read Operations
Fetch individual resources by ID (string):
Client.read("cus_123")
# => {:ok, %Req.Response{body: %{"id" => "cus_123", ...}}}
List resources by entity type (atom):
Client.read(:customers, limit: 10)
# => {:ok, %Req.Response{body: %{"object" => "list", "data" => [...]}}}
Create Operations
Create resources using entity type atoms:
Client.create(:customers, %{email: "user@example.com"})
# => {:ok, %Req.Response{body: %{"id" => "cus_new", ...}}}
Update Operations
Update resources by ID:
Client.update("cus_123", %{name: "New Name"})
# => {:ok, %Req.Response{body: %{"id" => "cus_123", "name" => "New Name"}}}
Delete Operations
Delete resources by ID:
Client.delete("cus_123")
# => {:ok, %Req.Response{body: %{"id" => "cus_123", "deleted" => true}}}
Supported Entity Types
The following entity types are supported (as atoms):
	:customers - Customer objects
	:products - Product objects
	:prices - Price objects
	:subscriptions - Subscription objects
	:invoices - Invoice objects
	:events - Event objects
	:checkout_sessions - Checkout Session objects

Error Handling
All operations return {:ok, response} or {:error, reason} tuples:
Client.read(:invalid_entity)
# => {:error, :unrecognized_entity_type}

# HTTP errors return the response with error details
Client.read("cus_nonexistent")
# => {:error, %Req.Response{status: 404, body: %{"error" => ...}}}
Bang Functions
Each CRUD function has a bang version (read!/2, create!/3, update!/3, delete!/2)
that raises a RuntimeError instead of returning error tuples:
# Raises on error
response = Client.read!("cus_123")
customer = response.body

# Raises RuntimeError: "Unrecognized entity type: :invalid"
Client.read!(:invalid)

# Raises RuntimeError: "Request failed with status 404: ..."
Client.read!("cus_nonexistent")
ID Prefix Recognition
The client automatically recognizes Stripe ID prefixes:
	cus_* → /customers/{id}
	product_* → /products/{id}
	price_* → /prices/{id}
	sub_* → /subscriptions/{id}
	inv_* → /invoices/{id}
	evt_* → /events/{id}
	cs_* → /checkout/sessions/{id}

Examples
# Fetch a customer
{:ok, response} = Client.read("cus_123")
customer = response.body

# List customers with pagination
{:ok, response} = Client.read(:customers, limit: 10, starting_after: "cus_123")
customers = response.body["data"]

# Create a customer
{:ok, response} = Client.create(:customers, %{
  email: "customer@example.com",
  name: "Jane Doe"
})

# Update a customer
{:ok, response} = Client.update("cus_123", %{
  metadata: %{user_id: "12345"}
})

# Delete a customer
{:ok, response} = Client.delete("cus_123")

    Functions
  


    
      
        create(entity, params, options \\ [])

      


        Creates a new resource.



    


    
      
        create!(entity, params, options \\ [])

      


        Creates a new resource, raising on error.



    


    
      
        delete(id, options \\ [])

      


        Deletes a resource by ID.



    


    
      
        delete!(id, options \\ [])

      


        Deletes a resource by ID, raising on error.



    


    
      
        entity_to_path(arg1)

      


        Converts an entity atom to its API path.



    


    
      
        new(options \\ [])

      


    


    
      
        parse_url(id)

      


        Parses a Stripe ID or URL path into a full API path.



    


    
      
        read(id_or_entity, options \\ [])

      


        Reads a resource by ID or lists resources by entity type.



    


    
      
        read!(id_or_entity, options \\ [])

      


        Reads a resource by ID or lists resources by entity type, raising on error.



    


    
      
        request(url, options \\ [])

      


    


    
      
        request!(url, options \\ [])

      


    


    
      
        update(id, params, options \\ [])

      


        Updates an existing resource by ID.



    


    
      
        update!(id, params, options \\ [])

      


        Updates an existing resource by ID, raising on error.

Creates a new resource.
Parameters
	entity - Atom representing the entity type (e.g., :customers, :products)
	params - Map of parameters for the resource
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_new", email: "new@example.com"})
...> end)
iex> {:ok, response} = PinStripe.Client.create(:customers, %{email: "new@example.com"})
iex> response.body["id"]
"cus_new"

# Unrecognized entity type
iex> PinStripe.Client.create(:invalid, %{})
{:error, :unrecognized_entity_type}

  



    

  
    
      
    
    
      create!(entity, params, options \\ [])



        
          
        

    

  


  

Creates a new resource, raising on error.
Similar to create/3 but raises a RuntimeError on error instead of returning an error tuple.
Parameters
	entity - Atom representing the entity type
	params - Map of parameters for the resource
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_new"})
...> end)
iex> response = PinStripe.Client.create!(:customers, %{email: "test@example.com"})
iex> response.body["id"]
"cus_new"

  



    

  
    
      
    
    
      delete(id, options \\ [])



        
          
        

    

  


  

Deletes a resource by ID.
Parameters
	id - String ID of the resource (e.g., "cus_123")
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", deleted: true})
...> end)
iex> {:ok, response} = PinStripe.Client.delete("cus_123")
iex> response.body["deleted"]
true

  



    

  
    
      
    
    
      delete!(id, options \\ [])



        
          
        

    

  


  

Deletes a resource by ID, raising on error.
Similar to delete/2 but raises a RuntimeError on error instead of returning an error tuple.
Parameters
	id - String ID of the resource
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", deleted: true})
...> end)
iex> response = PinStripe.Client.delete!("cus_123")
iex> response.body["deleted"]
true

Converts an entity atom to its API path.
Returns {:ok, path} for recognized entities, or {:error, :unrecognized_entity_type} otherwise.
Examples
iex> PinStripe.Client.entity_to_path(:customers)
{:ok, "/customers"}

iex> PinStripe.Client.entity_to_path(:products)
{:ok, "/products"}

iex> PinStripe.Client.entity_to_path(:invalid)
{:error, :unrecognized_entity_type}

  



    

  
    
      
    
    
      new(options \\ [])



        
          
        

    

  


  


  



  
    
      
    
    
      parse_url(id)



        
          
        

    

  


  

Parses a Stripe ID or URL path into a full API path.
Recognizes Stripe ID prefixes and converts them to the appropriate API endpoint.
Custom paths are returned as-is.
Examples
iex> PinStripe.Client.parse_url("cus_123")
"/customers/cus_123"

iex> PinStripe.Client.parse_url("product_abc")
"/products/product_abc"

iex> PinStripe.Client.parse_url("/custom/path")
"/custom/path"

  



    

  
    
      
    
    
      read(id_or_entity, options \\ [])



        
          
        

    

  


  

Reads a resource by ID or lists resources by entity type.
Parameters
	id_or_entity - Either a string ID (e.g., "cus_123") or an atom entity type (e.g., :customers)
	options - Keyword list of options (e.g., query parameters for list operations)

Examples
# Fetch by ID
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", email: "test@example.com"})
...> end)
iex> {:ok, response} = PinStripe.Client.read("cus_123")
iex> response.body["id"]
"cus_123"

# List resources
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{object: "list", data: []})
...> end)
iex> {:ok, response} = PinStripe.Client.read(:customers)
iex> response.body["object"]
"list"

# Unrecognized entity type
iex> PinStripe.Client.read(:invalid)
{:error, :unrecognized_entity_type}

  



    

  
    
      
    
    
      read!(id_or_entity, options \\ [])



        
          
        

    

  


  

Reads a resource by ID or lists resources by entity type, raising on error.
Similar to read/2 but raises a RuntimeError on error instead of returning an error tuple.
Parameters
	id_or_entity - Either a string ID or an atom entity type
	options - Keyword list of options

Examples
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123"})
...> end)
iex> response = PinStripe.Client.read!("cus_123")
iex> response.body["id"]
"cus_123"

  



    

  
    
      
    
    
      request(url, options \\ [])



        
          
        

    

  


  


  



    

  
    
      
    
    
      request!(url, options \\ [])



        
          
        

    

  


  


  



    

  
    
      
    
    
      update(id, params, options \\ [])



        
          
        

    

  


  

Updates an existing resource by ID.
Parameters
	id - String ID of the resource (e.g., "cus_123")
	params - Map of parameters to update
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", name: "Updated"})
...> end)
iex> {:ok, response} = PinStripe.Client.update("cus_123", %{name: "Updated"})
iex> response.body["name"]
"Updated"

  



    

  
    
      
    
    
      update!(id, params, options \\ [])



        
          
        

    

  


  

Updates an existing resource by ID, raising on error.
Similar to update/3 but raises a RuntimeError on error instead of returning an error tuple.
Parameters
	id - String ID of the resource
	params - Map of parameters to update
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(PinStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", name: "Updated"})
...> end)
iex> response = PinStripe.Client.update!("cus_123", %{name: "Updated"})
iex> response.body["name"]
"Updated"

  


        

      


  

    
PinStripe.WebhookController 
    



      
Base controller for handling Stripe webhook events.
This module provides a use macro that injects webhook handling functionality
into your Phoenix controller. It automatically verifies webhook signatures and
dispatches events to handler functions defined using the handle/2 macro.
Usage
Create a controller in your Phoenix app and define event handlers:
defmodule MyAppWeb.StripeWebhookController do
  use PinStripe.WebhookController

  handle "customer.created", fn event ->
    # Process customer.created event
    customer = event["data"]["object"]
    IO.inspect(customer, label: "New customer")
    :ok
  end

  handle "invoice.paid", MyApp.InvoicePaidHandler
end
Then add it to your router:
scope "/webhooks" do
  pipe_through [:api]

  post "/stripe", StripeWebhookController, :create
end
Configuration
Configure your webhook secret:
config :pin_stripe,
  stripe_webhook_secret: "whsec_..."
Security
This controller automatically verifies webhook signatures using the
stripe-signature header. Invalid signatures are rejected with a 400 response.
The raw request body must be available in conn.assigns.raw_body for signature
verification to work. Use PinStripe.ParsersWithRawBody in your endpoint.
Handler Functions
Handlers can be either:
	Anonymous functions that take the event as an argument
	Module names that implement a handle_event/1 function

Function Handler Example
handle "customer.created", fn event ->
  # Process customer.created event
  :ok
end
Module Handler Example
handle "invoice.paid", MyApp.InvoicePaidHandler
Then create the handler module:
defmodule MyApp.InvoicePaidHandler do
  def handle_event(event) do
    invoice = event["data"]["object"]
    # Process the paid invoice
    :ok
  end
end

      




  

    
PinStripe.WebhookHandler 
    



      
Spark DSL for defining webhook handlers.
This module provides a DSL for defining handlers for Stripe webhook events.
It is primarily used internally by PinStripe.WebhookController.
Example
defmodule MyApp.StripeWebhookHandlers do
  use PinStripe.WebhookHandler

  handle "customer.created", fn event ->
    # Process customer.created event
    :ok
  end

  handle "charge.succeeded", MyApp.ChargeHandler
end
Handlers can be either:
	Anonymous functions that take the event as an argument
	Module names that implement a handle_event/1 function

Options
	:extensions (list of module that adopts Spark.Dsl.Extension) - A list of DSL extensions to add to the Spark.Dsl

	:otp_app (atom/0) - The otp_app to use for any application configurable options

	:fragments (list of module/0) - Fragments to include in the Spark.Dsl. See the fragments guide for more.



      




  

    
PinStripe.WebhookSignature 
    



      
Verifies Stripe webhook signatures.
Stripe signs webhook events and includes the signature in the stripe-signature
header. This module verifies that signature to ensure the webhook came from Stripe.
See: https://docs.stripe.com/webhooks#verify-official-libraries
Configuration
Configure your webhook secret in your application config:
config :pin_stripe,
  stripe_webhook_secret: "whsec_..."
Usage
payload = ~s({"id": "evt_test", "type": "customer.created"})
signature = get_req_header(conn, "stripe-signature")
secret = Application.fetch_env!(:pin_stripe, :stripe_webhook_secret)

case WebhookSignature.verify(payload, signature, secret) do
  :ok ->
    # Process webhook
  {:error, reason} ->
    # Reject webhook
end

      


      
        Summary


  
    Functions
  


    
      
        sign(payload, timestamp, secret)

      


        Signs payload with timestamp and secret.



    


    
      
        verify(payload, signature, secret)

      


        Verifies payload against signature and secret.

          @spec sign(binary(), integer(), binary()) :: binary()


      


Signs payload with timestamp and secret.
Useful for testing webhook handling.
Examples
iex> payload = ~s({"type": "test"})
iex> timestamp = System.system_time(:second)
iex> secret = "whsec_test"
iex> signature = PinStripe.WebhookSignature.sign(payload, timestamp, secret)
iex> String.starts_with?(signature, "t=")
true

  



  
    
      
    
    
      verify(payload, signature, secret)



        
          
        

    

  


  

      

          @spec verify(binary(), binary(), binary()) :: :ok | {:error, binary()}


      


Verifies payload against signature and secret.
Returns :ok if the signature is valid and not expired.
Returns {:error, reason} otherwise.
The signature must be less than 300 seconds (5 minutes) old to prevent
replay attacks.
Examples
iex> payload = ~s({"type": "test"})
iex> timestamp = System.system_time(:second)
iex> secret = "whsec_test"
iex> signature = PinStripe.WebhookSignature.sign(payload, timestamp, secret)
iex> PinStripe.WebhookSignature.verify(payload, signature, secret)
:ok

  


        

      


  

    
PinStripe.Test.Fixtures 
    



      
Automatically generates and caches Stripe API fixtures for testing.
Fixtures are generated using your Stripe account's API version via the Stripe CLI
and cached in test/fixtures/stripe/. The API version is tracked in .api_version.
⚠️  Important: Side Effects
Fixture generation creates real test data in your Stripe account.
	Resources (customers, products, etc.) are created in test mode
	Objects are marked with "PinStripe Test Fixture" for identification
	Only test mode API keys (starting with sk_test_) are allowed
	Consider committing fixtures to git to avoid regenerating

Requirements
	Stripe CLI installed: brew install stripe/stripe-cli/stripe
	Stripe CLI authenticated: stripe login
	Test mode API key configured (starts with sk_test_)

Quick Start
Configuration
# config/test.exs
config :pin_stripe,
  stripe_api_key: System.get_env("STRIPE_SECRET_KEY")
Basic Usage
# In your test
test "creates a customer" do
  # Load/generate a customer fixture (use atoms for API resources)
  customer = PinStripe.Test.Fixtures.load(:customer)

  # Use with Req.Test
  Req.Test.stub(PinStripe, fn conn ->
    Req.Test.json(conn, customer)
  end)

  {:ok, response} = Client.create(:customers, %{email: "test@example.com"})
  assert response.body["object"] == "customer"
end
Customizing Fixtures
Generate fixtures with specific attributes by passing options:
# Customer with specific email (atoms for API resources)
customer = PinStripe.Test.Fixtures.load(:customer,
  email: "alice@example.com",
  name: "Alice Smith"
)

# Customer with metadata
customer = PinStripe.Test.Fixtures.load(:customer,
  metadata: %{user_id: "123", plan: "premium"}
)

# Webhook event with custom data (strings for webhook events - they have dots)
event = PinStripe.Test.Fixtures.load("customer.created",
  data: %{object: %{email: "custom@example.com"}}
)
Each unique set of options creates a separate cached fixture file.
The filename includes a hash of the options for uniqueness:
test/fixtures/stripe/
  customer.json              # Base customer
  customer-a3f2b9c1.json    # Customized customer
API Version Management
Fixtures match your Stripe account's default API version at the time they're
first generated. The version is stored in test/fixtures/stripe/.api_version.
When you upgrade your Stripe account's API version, run the sync task to clear
old fixtures and regenerate them with the new version:
mix pin_stripe.sync_api_version
This task:
	Detects your account's current API version
	Compares it with the cached version
	Clears all fixtures if the version changed
	Updates the .api_version file

Supported Fixtures
API Resources (Atoms)
	:customer - Customer objects
	:product - Product objects
	:price - Price objects
	:subscription - Subscription objects
	:invoice - Invoice objects
	:charge - Charge objects (one-time payments)
	:payment_intent - PaymentIntent objects (modern payment flow)
	:refund - Refund objects

Webhook Events (Strings - they have dots)
	"customer.created", "customer.updated", "customer.deleted"
	"customer.subscription.created", "customer.subscription.updated", "customer.subscription.deleted"
	"invoice.paid", "invoice.payment_failed"

Error Responses (Atoms)
	:error_404 - Not found (resource_missing)
	:error_400 - Bad request (invalid_request_error)
	:error_401 - Unauthorized (authentication error)
	error_429 - Rate limit exceeded

Testing Best Practices
Commit Fixtures to Git
To avoid generating fixtures in CI or on every developer machine:
# Commit fixtures to version control
git add test/fixtures/stripe
git commit -m "Add Stripe test fixtures"
Use Base Fixtures and Modify
Instead of generating many custom fixtures:
# Load base fixture
customer = PinStripe.Test.Fixtures.load(:customer)

# Modify as needed in your test
customer = Map.put(customer, "email", "specific@test.com")
Clean Up Test Data
Objects created during fixture generation accumulate in your Stripe test account.
They're marked with pinstripe_fixture: true metadata for easy identification.
To clean up manually:
# List PinStripe test objects
stripe customers list --limit 100 | grep "PinStripe Test Fixture"

# Delete a specific customer
stripe customers delete cus_xxx
Examples
Testing API Responses
test "handles customer creation" do
  fixture = PinStripe.Test.Fixtures.load(:customer)

  Req.Test.stub(PinStripe, fn conn ->
    Req.Test.json(conn, fixture)
  end)

  {:ok, response} = Client.create(:customers, %{email: "test@example.com"})
  assert response.body["id"] == fixture["id"]
end
Testing Webhook Handlers
test "handles customer.created webhook" do
  event = PinStripe.Test.Fixtures.load("customer.created")

  conn = build_webhook_conn(event)
  conn = MyAppWeb.StripeWebhookController.create(conn, event)

  assert conn.status == 200
end
Testing Error Handling
test "handles not found error" do
  error = PinStripe.Test.Fixtures.load(:error_404)

  Req.Test.stub(PinStripe, fn conn ->
    conn
    |> Plug.Conn.put_status(404)
    |> Req.Test.json(error)
  end)

  assert {:error, %{status: 404}} = Client.read("cus_nonexistent")
end
Testing with Multiple Customers
test "handles multiple customers" do
  alice = PinStripe.Test.Fixtures.load(:customer, email: "alice@test.com")
  bob = PinStripe.Test.Fixtures.load(:customer, email: "bob@test.com")

  # Each gets cached separately
  assert alice["email"] == "alice@test.com"
  assert bob["email"] == "bob@test.com"
end

    Functions
  


    
      
        api_version()

      


        Returns the API version used for cached fixtures.



    


    
      
        detect_account_api_version()

      


        Detects the current Stripe account's API version.



    


    
      
        list()

      


        Lists all available (cached) fixtures.



    


    
      
        load(fixture_name, opts \\ [])

      


        Loads a fixture by name, generating it if not cached.



    





      


      
        Functions


        


  
    
      
    
    
      api_version()



        
          
        

    

  


  

Returns the API version used for cached fixtures.
Returns nil if no .api_version file exists yet.
Examples
iex> # When no .api_version file exists
iex> Application.put_env(:pin_stripe, :fixtures_dir, "test/fixtures/doctest_no_version")
iex> File.rm_rf!("test/fixtures/doctest_no_version")
iex> PinStripe.Test.Fixtures.api_version()
nil

iex> # After creating a version file
iex> Application.put_env(:pin_stripe, :fixtures_dir, "test/fixtures/doctest_version")
iex> File.rm_rf!("test/fixtures/doctest_version")
iex> File.mkdir_p!("test/fixtures/doctest_version")
iex> File.write!("test/fixtures/doctest_version/.api_version", "2024-06-20")
iex> PinStripe.Test.Fixtures.api_version()
"2024-06-20"
iex> File.rm_rf!("test/fixtures/doctest_version")
iex> Application.delete_env(:pin_stripe, :fixtures_dir)
:ok

  



  
    
      
    
    
      detect_account_api_version()



        
          
        

    

  


  

Detects the current Stripe account's API version.
Makes a quick API call to determine the version. Requires a valid Stripe API key
and authenticated Stripe CLI.
Examples
# Requires real Stripe setup
PinStripe.Test.Fixtures.detect_account_api_version()
#=> "2024-06-20"

  



  
    
      
    
    
      list()



        
          
        

    

  


  

Lists all available (cached) fixtures.
Returns a list of fixture filenames (without .json extension), sorted alphabetically.
Returns an empty list if no fixtures directory exists.
Examples
iex> # When no fixtures exist
iex> Application.put_env(:pin_stripe, :fixtures_dir, "test/fixtures/doctest_empty")
iex> File.rm_rf!("test/fixtures/doctest_empty")
iex> PinStripe.Test.Fixtures.list()
[]

iex> # Lists cached API resource fixtures (not error fixtures which use atoms)
iex> # Error fixtures don't create files, so list() returns only cached API resources
iex> Application.put_env(:pin_stripe, :fixtures_dir, "test/fixtures/stripe")
iex> list = PinStripe.Test.Fixtures.list()
iex> "error_400" in list
true
iex> "error_404" in list
true
iex> Application.delete_env(:pin_stripe, :fixtures_dir)
:ok

  



    

  
    
      
    
    
      load(fixture_name, opts \\ [])



        
          
        

    

  


  

Loads a fixture by name, generating it if not cached.
Fixture Types
Error Fixtures (Atoms) - Self-contained, no Stripe CLI required:
	Use atoms: :error_400, :error_401, :error_402, :error_403, :error_404, 
:error_409, :error_424, :error_429, :error_500, :error_502, 
:error_503, :error_504
	Generated instantly (no API calls)
	Match actual Stripe error responses
	Not cached to filesystem

API Resources (Atoms) - Require Stripe CLI and API key:
	Use atoms: :customer, :product, :price, :subscription, 
:invoice, :charge, :payment_intent, :refund
	Created via Stripe API
	Cached to filesystem
	Options passed to Stripe CLI during creation

Webhook Events (Strings) - Require Stripe CLI and API key:
	Use strings with dots: "customer.created", "invoice.paid", etc.
	Generated via Stripe CLI trigger
	Cached to filesystem
	Options used to modify event data after generation

Options
Options are used to customize fixtures. Each unique combination of options 
creates a separate cached fixture (for API resources/webhooks only).
Examples
Error fixtures use atoms and are self-contained:
iex> error = PinStripe.Test.Fixtures.load(:error_404)
iex> error["error"]["type"]
"invalid_request_error"
iex> error["error"]["code"]
"resource_missing"

iex> error = PinStripe.Test.Fixtures.load(:error_400)
iex> error["error"]["type"]
"invalid_request_error"
iex> error["error"]["code"]
"parameter_invalid_empty"

iex> error = PinStripe.Test.Fixtures.load(:error_401)
iex> error["error"]["type"]
"invalid_request_error"

iex> error = PinStripe.Test.Fixtures.load(:error_429)
iex> error["error"]["type"]
"rate_limit_error"
API resources use atoms and webhook events use strings (both require Stripe CLI):
# API resources use atoms - require real Stripe setup
PinStripe.Test.Fixtures.load(:customer)
#=> %{"id" => "cus_...", "object" => "customer", ...}

PinStripe.Test.Fixtures.load(:customer, email: "test@example.com")
#=> %{"id" => "cus_...", "email" => "test@example.com", ...}

# Webhook events use strings (they have dots)
PinStripe.Test.Fixtures.load("customer.created")
#=> %{"id" => "evt_...", "type" => "customer.created", ...}

  


        

      


  

    
PinStripe.Test.Mock 
    



      
Test mocking utilities for PinStripe, wrapping Req.Test with sensible defaults.
This module provides a thin wrapper around Req.Test with two main benefits:
	Single Namespace: All PinStripe testing utilities under PinStripe.Test.*
	Sensible Defaults: Functions default to using PinStripe as the stub name

Relationship to Req.Test
This module delegates core functionality to Req.Test. For advanced use cases or
detailed documentation, refer to the Req.Test documentation.
The main functions are:
	stub/1 and stub/2 - Create request stubs
	expect/1, expect/2, and expect/3 - Create request expectations
	allow/2 and allow/3 - Allow other processes to use stubs/mocks
	json/2 - Create JSON responses (Stripe API returns JSON)
	transport_error/2 - Simulate network errors

High-Level Stripe Helpers
For common Stripe operations, use these helpers that automatically handle URL resolution:
	stub_read/2 - Stub read/list operations
	stub_create/2 - Stub create operations
	stub_update/2 - Stub update operations
	stub_delete/2 - Stub delete operations
	stub_error/3 - Stub error responses

Setup
Configure your test environment to use Req.Test:
# config/test.exs
config :pin_stripe,
  stripe_api_key: "sk_test_...",
  req_options: [plug: {Req.Test, PinStripe}]
High-Level Helpers (Recommended)
The easiest way to stub PinStripe operations is to use the high-level helper functions
that mirror PinStripe.Client operations. These automatically handle URL resolution
and HTTP method matching:
test "reads a customer using stub_read" do
  Mock.stub_read("cus_123", %{"id" => "cus_123", "email" => "test@example.com"})

  {:ok, response} = Client.read("cus_123")
  assert response.body["email"] == "test@example.com"
end

test "creates a customer using stub_create" do
  Mock.stub_create(:customers, %{"id" => "cus_new", "email" => "new@example.com"})

  {:ok, response} = Client.create(:customers, %{email: "new@example.com"})
  assert response.body["id"] == "cus_new"
end

test "updates a customer using stub_update" do
  Mock.stub_update("cus_123", %{"id" => "cus_123", "name" => "Updated Name"})

  {:ok, response} = Client.update("cus_123", %{name: "Updated Name"})
  assert response.body["name"] == "Updated Name"
end

test "deletes a customer using stub_delete" do
  Mock.stub_delete("cus_123", %{"id" => "cus_123", "deleted" => true})

  {:ok, response} = Client.delete("cus_123")
  assert response.body["deleted"] == true
end

test "stubs errors using stub_error" do
  Mock.stub_error("cus_nonexistent", 404, %{
    "error" => %{"code" => "resource_missing"}
  })

  {:error, response} = Client.read("cus_nonexistent")
  assert response.status == 404
end
These helpers work seamlessly with fixtures:
test "uses fixtures with helpers" do
  customer = Fixtures.load(:customer)
  Mock.stub_read("cus_123", customer)

  error = Fixtures.load(:error_404)
  Mock.stub_error("cus_missing", 404, error)
end
Basic Usage with Raw Data (Lower-Level)
The simplest approach is to stub responses with inline data:
test "reads a customer" do
  Mock.stub(fn conn ->
    if conn.method == "GET" and conn.request_path == "/v1/customers/cus_123" do
      Mock.json(conn, %{"id" => "cus_123", "email" => "test@example.com"})
    else
      conn
    end
  end)

  {:ok, response} = Client.read("cus_123")
  assert response.body["email"] == "test@example.com"
end

test "creates a customer" do
  Mock.stub(fn conn ->
    if conn.method == "POST" and conn.request_path == "/v1/customers" do
      Mock.json(conn, %{"id" => "cus_new", "email" => "new@example.com"})
    else
      conn
    end
  end)

  {:ok, response} = Client.create(:customers, %{email: "new@example.com"})
  assert response.body["id"] == "cus_new"
end

test "lists customers" do
  Mock.stub(fn conn ->
    if conn.method == "GET" and conn.request_path == "/v1/customers" do
      Mock.json(conn, %{
        "object" => "list",
        "data" => [
          %{"id" => "cus_1", "email" => "user1@example.com"},
          %{"id" => "cus_2", "email" => "user2@example.com"}
        ],
        "has_more" => false
      })
    else
      conn
    end
  end)

  {:ok, response} = Client.read(:customers)
  assert length(response.body["data"]) == 2
end
Using Fixtures
For more realistic test data, use PinStripe.Test.Fixtures to load fixture files:
test "reads a customer using fixture" do
  Mock.stub(fn conn ->
    if conn.method == "GET" and conn.request_path == "/v1/customers/cus_123" do
      customer = Fixtures.load(:customer)
      Mock.json(conn, customer)
    else
      conn
    end
  end)

  {:ok, response} = Client.read("cus_123")
  assert response.body["object"] == "customer"
end

test "reads a customer with custom fixture data" do
  Mock.stub(fn conn ->
    if conn.method == "GET" and conn.request_path == "/v1/customers/cus_123" do
      customer = Fixtures.load(:customer, email: "custom@example.com", name: "Custom Name")
      Mock.json(conn, customer)
    else
      conn
    end
  end)

  {:ok, response} = Client.read("cus_123")
  assert response.body["email"] == "custom@example.com"
  assert response.body["name"] == "Custom Name"
end
Stubbing Errors
Use Plug.Conn.put_status/2 to return error responses:
test "handles 404 error" do
  Mock.stub(fn conn ->
    if conn.method == "GET" and conn.request_path == "/v1/customers/cus_nonexistent" do
      conn
      |> Plug.Conn.put_status(404)
      |> Mock.json(%{
        "error" => %{
          "type" => "invalid_request_error",
          "code" => "resource_missing",
          "message" => "No such customer: cus_nonexistent"
        }
      })
    else
      conn
    end
  end)

  assert {:error, %{status: 404}} = Client.read("cus_nonexistent")
end

test "handles error using fixture" do
  Mock.stub(fn conn ->
    if conn.method == "POST" and conn.request_path == "/v1/customers" do
      error = Fixtures.load(:error_400)
      conn
      |> Plug.Conn.put_status(400)
      |> Mock.json(error)
    else
      conn
    end
  end)

  {:error, response} = Client.create(:customers, %{})
  assert response.status == 400
  assert response.body["error"]["code"] == "parameter_invalid_empty"
end
Pattern Matching on Multiple Requests
You can handle multiple request patterns in a single stub:
test "handles multiple operations" do
  Mock.stub(fn conn ->
    case {conn.method, conn.request_path} do
      {"GET", "/v1/customers/" <> id} ->
        Mock.json(conn, %{"id" => id, "email" => "#{id}@example.com"})

      {"POST", "/v1/customers"} ->
        Mock.json(conn, %{"id" => "cus_new", "email" => "new@example.com"})

      {"DELETE", "/v1/customers/" <> id} ->
        Mock.json(conn, %{"id" => id, "deleted" => true, "object" => "customer"})

      _ ->
        conn
    end
  end)

  {:ok, read_response} = Client.read("cus_123")
  assert read_response.body["id"] == "cus_123"

  {:ok, create_response} = Client.create(:customers, %{email: "new@example.com"})
  assert create_response.body["id"] == "cus_new"

  {:ok, delete_response} = Client.delete("cus_123")
  assert delete_response.body["deleted"] == true
end
Available Fixtures
See PinStripe.Test.Fixtures for the full list of available fixtures:
	:customer - A Stripe customer object
	:charge - A Stripe charge object
	:payment_intent - A Stripe payment intent object
	:refund - A Stripe refund object
	:error_400, :error_401, :error_403, :error_404, etc. - Various error responses

All fixtures can be customized with keyword options that override specific fields.

        allow(owner, pid)

      


        Allows the process pid to use stubs defined by owner.



    


    
      
        allow(name, owner, pid)

      


        Allows the process pid to use stubs defined by owner for the given name.



    


    
      
        expect(plug)

      


        Creates a request expectation expected to be called once.



    


    
      
        expect(plug, count)

      


        Creates a request expectation with the given plug and count.



    


    
      
        expect(plug, count, name)

      


        Creates a request expectation with the given plug, count, and name.



    


    
      
        json(conn, data)

      


        Sends a JSON response. See Req.Test.json/2.



    


    
      
        stub(plug)

      


        Creates a request stub with the given plug.



    


    
      
        stub(plug, name)

      


        Creates a request stub with the given plug and custom name.

        Stubs a create operation for the given entity type.



    


    
      
        stub_delete(id, response_data)

      


        Stubs a delete operation for the given ID.



    


    
      
        stub_error(id, status, error_data)

      


        Stubs an error response for the given ID, entity type, or any request.



    


    
      
        stub_read(id_or_entity, response_data)

      


        Stubs a read operation for the given ID or entity type.



    


    
      
        stub_update(id, response_data)

      


        Stubs an update operation for the given ID.



    


    
      
        transport_error(conn, reason)

      


        Simulates a network transport error. See Req.Test.transport_error/2.

Allows the process pid to use stubs defined by owner.
Defaults to using PinStripe as the stub name. For custom names, use allow/3.
See Req.Test.allow/3 for full documentation on allowances and concurrent testing.
Examples
test "allows spawned process to use stubs" do
  {:ok, pid} = start_supervised(MyWorker)

  Mock.stub(fn conn ->
    Mock.json(conn, %{id: "test"})
  end)

  Mock.allow(self(), pid)

  # Now pid can use the stub
end

  



  
    
      
    
    
      allow(name, owner, pid)



        
          
        

    

  


  

Allows the process pid to use stubs defined by owner for the given name.
See Req.Test.allow/3 for full documentation.

  



  
    
      
    
    
      expect(plug)



        
          
        

    

  


  

Creates a request expectation expected to be called once.
Defaults to using PinStripe as the stub name. For custom counts or names, use expect/2 or expect/3.
Examples
test "expects one request" do
  Mock.expect(fn conn ->
    Mock.json(conn, %{id: "cus_123"})
  end)

  {:ok, _} = Client.read("cus_123")
end

  



  
    
      
    
    
      expect(plug, count)



        
          
        

    

  


  

Creates a request expectation with the given plug and count.
Defaults to using PinStripe as the stub name. For custom names, use expect/3.
Examples
test "expects multiple requests" do
  Mock.expect(fn conn ->
    Mock.json(conn, %{id: "test"})
  end, 2)

  {:ok, _} = Client.read("test_123")
  {:ok, _} = Client.read("test_456")
end

  



  
    
      
    
    
      expect(plug, count, name)



        
          
        

    

  


  

Creates a request expectation with the given plug, count, and name.
See Req.Test.expect/3 for full documentation.
Examples
test "expects requests with custom name" do
  Mock.expect(fn conn ->
    Mock.json(conn, %{id: "test"})
  end, 2, :custom)
end

Sends a JSON response. See Req.Test.json/2.
The Stripe API returns JSON for all responses, so this is the primary
response helper you'll use in tests.

Creates a request stub with the given plug.
Defaults to using PinStripe as the stub name. For custom names, use stub/2.
Examples
iex> alias PinStripe.Test.Mock
iex> Mock.stub(fn conn ->
...>   Mock.json(conn, %{"id" => "cus_123"})
...> end)
:ok
iex> {:ok, response} = PinStripe.Client.read("cus_123")
iex> response.body["id"]
"cus_123"

  



  
    
      
    
    
      stub(plug, name)



        
          
        

    

  


  

Creates a request stub with the given plug and custom name.
See Req.Test.stub/2 for full documentation.
Examples
test "stubs with custom name" do
  Mock.stub(fn conn ->
    Mock.json(conn, %{id: "test"})
  end, :custom_name)
end

Stubs a create operation for the given entity type.
This is a convenience wrapper around stub/1 that automatically handles:
	Entity type to path conversion via PinStripe.Client.entity_to_path/1
	Proper HTTP method matching (POST)

Parameters
	entity - Atom representing the entity type (e.g., :customers, :products)
	response_data - Map of data to return in the JSON response

Examples
# Stub creating a customer
iex> PinStripe.Test.Mock.stub_create(:customers, %{"id" => "cus_new", "email" => "new@example.com"})
:ok
iex> {:ok, response} = PinStripe.Client.create(:customers, %{email: "new@example.com"})
iex> response.body["id"]
"cus_new"

# Stub creating a product
iex> PinStripe.Test.Mock.stub_create(:products, %{"id" => "product_new", "name" => "Widget"})
:ok
iex> {:ok, response} = PinStripe.Client.create(:products, %{name: "Widget"})
iex> response.body["name"]
"Widget"

Stubs a delete operation for the given ID.
This is a convenience wrapper around stub/1 that automatically handles:
	URL resolution via PinStripe.Client.parse_url/1
	Proper HTTP method matching (DELETE)

Parameters
	id - String ID of the resource (e.g., "cus_123")
	response_data - Map of data to return in the JSON response (typically includes "deleted" => true)

Examples
# Stub deleting a customer
iex> PinStripe.Test.Mock.stub_delete("cus_123", %{
...>   "id" => "cus_123",
...>   "deleted" => true,
...>   "object" => "customer"
...> })
:ok
iex> {:ok, response} = PinStripe.Client.delete("cus_123")
iex> response.body["deleted"]
true

# Stub deleting a product
iex> PinStripe.Test.Mock.stub_delete("product_abc", %{"id" => "product_abc", "deleted" => true})
:ok
iex> {:ok, response} = PinStripe.Client.delete("product_abc")
iex> response.body["deleted"]
true

  



  
    
      
    
    
      stub_error(id, status, error_data)



        
          
        

    

  


  

Stubs an error response for the given ID, entity type, or any request.
This is a convenience wrapper around stub/1 that automatically handles:
	URL resolution for string IDs via PinStripe.Client.parse_url/1
	Entity type to path conversion for atom entity types
	Setting the appropriate HTTP status code
	Matching any request when :any is passed

Parameters
	id_entity_or_any - Either a string ID (e.g., "cus_123"), an atom entity type (e.g., :customers), or :any to match all requests
	status - HTTP status code (e.g., 404, 400, 401)
	error_data - Map of error data to return in the JSON response

Examples
# Stub a 404 error for a specific customer
iex> PinStripe.Test.Mock.stub_error("cus_nonexistent", 404, %{
...>   "error" => %{
...>     "type" => "invalid_request_error",
...>     "code" => "resource_missing"
...>   }
...> })
:ok
iex> {:error, response} = PinStripe.Client.read("cus_nonexistent")
iex> response.status
404

# Stub a 400 error for customer creation
iex> PinStripe.Test.Mock.stub_error(:customers, 400, %{
...>   "error" => %{"code" => "parameter_invalid_empty"}
...> })
:ok
iex> {:error, response} = PinStripe.Client.create(:customers, %{})
iex> response.status
400

# Stub an error for any request
iex> PinStripe.Test.Mock.stub_error(:any, 401, %{"error" => %{"message" => "Invalid API key"}})
:ok
iex> {:error, response} = PinStripe.Client.read("cus_123")
iex> response.status
401

  



  
    
      
    
    
      stub_read(id_or_entity, response_data)



        
          
        

    

  


  

Stubs a read operation for the given ID or entity type.
This is a convenience wrapper around stub/1 that automatically handles:
	URL resolution via PinStripe.Client.parse_url/1 (for string IDs)
	Entity type to path conversion via PinStripe.Client.entity_to_path/1 (for atom entity types)
	Proper HTTP method matching (GET)

Parameters
	id_or_entity - Either a string ID (e.g., "cus_123") or an atom entity type (e.g., :customers)
	response_data - Map of data to return in the JSON response

Examples
# Stub reading a customer by ID
iex> PinStripe.Test.Mock.stub_read("cus_123", %{"id" => "cus_123", "email" => "test@example.com"})
:ok
iex> {:ok, response} = PinStripe.Client.read("cus_123")
iex> response.body["email"]
"test@example.com"

# Stub reading a product by ID
iex> PinStripe.Test.Mock.stub_read("product_abc", %{"id" => "product_abc", "name" => "Widget"})
:ok
iex> {:ok, response} = PinStripe.Client.read("product_abc")
iex> response.body["name"]
"Widget"

# Stub listing customers by entity type
iex> PinStripe.Test.Mock.stub_read(:customers, %{
...>   "object" => "list",
...>   "data" => [%{"id" => "cus_1"}],
...>   "has_more" => false
...> })
:ok
iex> {:ok, response} = PinStripe.Client.read(:customers)
iex> response.body["object"]
"list"

Stubs an update operation for the given ID.
This is a convenience wrapper around stub/1 that automatically handles:
	URL resolution via PinStripe.Client.parse_url/1
	Proper HTTP method matching (POST for updates in Stripe API)

Parameters
	id - String ID of the resource (e.g., "cus_123")
	response_data - Map of data to return in the JSON response

Examples
# Stub updating a customer
iex> PinStripe.Test.Mock.stub_update("cus_123", %{"id" => "cus_123", "name" => "Updated Name"})
:ok
iex> {:ok, response} = PinStripe.Client.update("cus_123", %{name: "Updated Name"})
iex> response.body["name"]
"Updated Name"

# Stub updating a subscription
iex> PinStripe.Test.Mock.stub_update("sub_xyz", %{"id" => "sub_xyz", "status" => "canceled"})
:ok
iex> {:ok, response} = PinStripe.Client.update("sub_xyz", %{status: "canceled"})
iex> response.body["status"]
"canceled"

  



  
    
      
    
    
      transport_error(conn, reason)



        
          
        

    

  


  

Simulates a network transport error. See Req.Test.transport_error/2.
Useful for testing network failure scenarios like connection timeouts or refused connections.

  


        

      


  

    
PinStripe.ParsersWithRawBody 
    



      
A custom Plug.Parsers that caches the raw request body for webhook routes.
This plug conditionally caches the raw body only for webhook endpoints
configured in your application. For all other routes, it behaves like standard
Plug.Parsers.
The raw body is needed for webhook signature verification, as the signature
is computed over the exact bytes received from Stripe.
Usage
In your Phoenix endpoint, replace Plug.Parsers with this module:
# Before:
plug Plug.Parsers,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Phoenix.json_library()

# After:
plug PinStripe.ParsersWithRawBody,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Phoenix.json_library()
The raw body will be available in conn.assigns.raw_body as a list of
binary chunks for webhook routes.
Configuration
Configure webhook paths in your config/runtime.exs as a list:
config :pin_stripe,
  webhook_paths: ["/webhooks/stripe"]
You can configure multiple webhook endpoints:
config :pin_stripe,
  webhook_paths: ["/webhooks/stripe", "/webhooks/stripe_connect"]
If no configuration is provided, the default path ["/webhooks/stripe"] will be used.
Multiple Webhook Endpoints
To add additional webhook endpoints:
	Add the path to the :webhook_paths config (as shown above)

	Create a new controller that uses PinStripe.WebhookController:
 defmodule MyAppWeb.StripeConnectWebhookController do
   use PinStripe.WebhookController
   handle "account.updated", fn event ->
 # Handle Connect events
 :ok
   end
 end

	Add the route in your router:
 scope "/webhooks" do
   post "/stripe_connect", MyAppWeb.StripeConnectWebhookController, :create
 end

        cache_raw_body(conn, opts)

      


        Custom body reader that caches the raw request body.

Custom body reader that caches the raw request body.
This function is passed to Plug.Parsers via the :body_reader option.
It reads the body and stores it in conn.assigns.raw_body as a list
of chunks (prepended for efficiency).

  


        

      


  

    
PinStripe.WebhookHandler.Dsl 
    



      
DSL extension for defining webhook handlers.

      




  

    
PinStripe.WebhookHandler.Info 
    



      
Introspection functions for webhook handler DSL.

        handlers DSL entities

          @spec handlers(dsl_or_extended :: module() | map()) :: [struct()]


      


handlers DSL entities

  


        

      


  

    
mix pin_stripe.gen.handler 
    



      
Generate a Stripe webhook event handler
Generates a handler for a specific Stripe webhook event.
Example
mix pin_stripe.gen.handler customer.created

Arguments
	event - The Stripe event name (e.g., "customer.created", "charge.succeeded")

Options
	--handler-type - Type of handler to generate: "function" or "module" (default: prompts user)
	--module - Custom module name for module handlers (default: auto-generated from event name)
	--create-handler-module - Module name to create if no WebhookHandler module exists

 ## Handler Types
 ### Function Handler
 Generates an inline function handler in your WebhookController:
 handle "customer.created", fn event ->
   # Handle customer.created event
   :ok
 end
 ### Module Handler
 Generates a separate module with a handle_event/1 function:
 # In your WebhookController
 handle "customer.created", MyAppWeb.StripeWebhookHandlers.CustomerCreated

 # Generated module at lib/my_app_web/stripe_webhook_handlers/customer_created.ex
 defmodule MyAppWeb.StripeWebhookHandlers.CustomerCreated do
   def handle_event(event) do
     # Handle customer.created event
     :ok
   end
 end

      




  

    
mix pin_stripe.install 
    



      
Install PinStripe with Stripe webhook support
This installer will:
	Replace Plug.Parsers with PinStripe.ParsersWithRawBody in your Phoenix endpoint
	Generate a StripeWebhookController in lib/{app}_web with example event handlers
	Add a webhook route to your router that points to the generated controller
	Configure the webhook_paths in config/runtime.exs
	Add :pin_stripe to import_deps in .formatter.exs for DSL formatting support

The ParsersWithRawBody plug caches the raw request body for webhook signature verification,
as required by Stripe's webhook security. The webhook path is configurable and supports
multiple endpoints.
The generated controller automatically handles signature verification and dispatches events
to handler functions you define using the handle DSL.
Example
mix pin_stripe.install --path /webhooks/stripe

Options
	--path or -p - The webhook endpoint path (default: "/webhooks/stripe")


      




  

    
mix pin_stripe.sync_api_version 
    



      
Detects your Stripe account's API version and updates fixtures accordingly.
When your Stripe account upgrades to a new API version, this task will:
	Detect the new API version from your account
	Compare with the cached version in .api_version
	If different, update .api_version and clear all fixtures
	Fixtures will regenerate with the new version on next test run

Examples
# Sync API version
mix pin_stripe.sync_api_version
Requirements
	Stripe CLI installed and authenticated
	Test mode API key configured


      




  

    
mix pin_stripe.sync_webhook_handlers 
    



      
Sync webhook handlers with Stripe's configured webhook endpoints
This task will:
	Fetch all webhook endpoints from your Stripe account using the Stripe CLI
	Extract all enabled events from those endpoints
	Compare them with handlers in your WebhookHandler module
	Generate handlers for any missing events

This is useful for keeping your local webhook handlers in sync with your
Stripe webhook configuration, especially after adding new events in the
Stripe Dashboard.
Example
mix pin_stripe.sync_webhook_handlers

Options
	--api-key or -k - The Stripe API key to use (prompts to use config key if not provided)
	--handler-type or -t - Type of handler to generate: "function", "module", or "ask" (default: prompts user)
	--skip-confirmation or -y - Skip confirmation prompts and generate all missing handlers
	--create-handler-module - Module name to create if no WebhookHandler module exists

Prerequisites
This task requires the Stripe CLI to be installed and authenticated:
# Install Stripe CLI (macOS)
brew install stripe/stripe-cli/stripe

# Login to Stripe
stripe login

Handler Types
Function Handler (inline)
Generates inline function handlers in your WebhookHandler module:
handle "customer.created", fn event ->
  # Handle customer.created event
  :ok
end
Module Handler
Generates separate modules with a handle_event/1 function:
# In your WebhookHandler module
handle "customer.created", MyApp.StripeWebhookHandlers.CustomerCreated

# Generated module
defmodule MyApp.StripeWebhookHandlers.CustomerCreated do
  def handle_event(event) do
    # Handle customer.created event
    :ok
  end
end
Ask (interactive)
Prompts you for each event individually, allowing you to choose
function or module handlers on a per-event basis.

      




  

    
mix pin_stripe.update_supported_events 
    



      
Updates the list of supported Stripe webhook events.
This task is for library contributors to update the list of valid Stripe events
that can be used with the webhook handler generators.
Usage
mix pin_stripe.update_supported_events
This task will:
	Run stripe trigger --help to get the list of supported events
	Parse the output to extract event names
	Write the events to priv/supported_stripe_events.txt

Requirements
You must have the Stripe CLI installed and available in your PATH.
Install it from: https://stripe.com/docs/stripe-cli

